微算機實驗報告

Lab # 11

姓名：仇健安
系級：電機系
學號：111511239
上課時間：2025/05/20

一、實驗目的：

本實驗透過 8051 微控制器結合矩陣式鍵盤輸入，實現音階控制與蜂鳴器驅動的功能。透過對鍵盤掃描與按鍵判斷的實作，取代原本老師還沒上到的 serial input，並將其與 Timer 計時器結合，用以控制蜂鳴器輸出不同頻率的方波，對應 Do 到 Si 七個音階。

二、硬體架構：



蜂鳴器腳位	**連接對象**	**說明**
VCC（+）	+5V	可接開發板 5V 或模組自帶電源
GND（–）	GND	接開發板的 GND 共地
控制腳（IN）	**P0.0**	接開發板 P0.0 腳，輸出方波
鍵盤腳位	**連接對象**	**說明**
Row 1	**P1.7**	掃描鍵盤列的第 1 列
Row 2	**P1.6**	掃描鍵盤列的第 2 列
Row 3	**P1.5**	掃描鍵盤列的第 3 列（可略）
Row 4	**P1.4**	掃描鍵盤列的第 4 列
Column 1	**P1.3**	掃描鍵盤行的第 1 行
Column 2	**P1.2**	掃描鍵盤行的第 2 行
Column 3	**P1.1**	掃描鍵盤行的第 3 行
Column 4	**P1.0**	掃描鍵盤行的第 4 行

三、程式流程圖：

鍵盤部分，因為只用到 8 個按鍵(我直接把加分題做在一起，所以會用到除了 DO~SI 之外還有播歌總共八個按鍵)，所以只需要掃描 ROW1、ROW2:



判斷該撥放的音頻或播歌的指令流程，因為只有掃描 8 個按鍵，且按下按鈕才會來這邊，所以 SONG 那邊不用再去進行判斷，只要前面七個不是，一定是播歌:



播歌的程式流程圖，根據 TABLE 的值，判斷要撥放什麼音頻:

播放音頻的程式流程圖，因為 DO~SI 都一樣，只是頻率不同所以 TIMER 的數值設定不一樣即可，這邊只畫出 DO 的播放流程圖:



四、問題與討論：

(1) 試根據你所設計的程式架構，解說如何實現蜂鳴器的長短音功能。

我的設計中，蜂鳴器是透過 8051 的 Timer0 產生固定頻率的方波，並由 P0.0 腳輸出控制蜂鳴器發聲。聲音的長短是由兩層迴圈控制的：

- R2：內層迴圈，控制每次產生方波的次數。
- R3：外層迴圈，控制整體音符持續的時間。

以播放 Do 音為例，程式架構如下：

```
MOV R3, #3          ; 外圈次數，控制音長
PDLOOP:
  MOV R2, #130      ; 內圈次數，控制翻轉頻率
PDLOOP1:
  ACALL INVERTDO
  DJNZ R2, PDLOOP1
  DJNZ R3, PDLOOP
```

只要修改 R2 或 R3 的數值，即可改變聲音的長短。數值越大，聲音持續越久；數值越小，聲音越短。

(2) 假設要更改音色，試問要如何更改?

蜂鳴器的音色主要取決於所輸出的波形類型與頻率。目前程式透過 CPL P0.0 搭配 Timer0 所產生的是標準方波，因此是透過改變頻率更改音色，因此方法如下：

調整頻率：

透過修改 Timer0 的 TH0 和 TL0 設定值，可以調整方波頻率，從而改變音符的高低，進而讓人聽感上的音色有變化。

五、程式碼與註解：

```
ORG 0000H
AJMP INITIAL
ORG 0050H

;======================
; 初始化設定
;======================
INITIAL:
    MOV TMOD, #00000000B    ; Timer0 設定為 Mode 0（13-bit），掃描鍵盤與控制頻率用
    SETB P0.0               ; 預設將蜂鳴器腳位設為高電位，避免初始發聲

;======================
; 鍵盤掃描主迴圈
;======================
ROW1:
    MOV P1, #7FH            ; 掃描第 1 列（P1.7 = 0）
    LCALL DELAY             ; 延遲去彈跳
    MOV A, P1
```

```
    ANL A, #0FH              ; 擷取 P1.3~P1.0（讀取行）
    MOV R0, #0               ; 記錄第 1 列的 base index 為 0
    CJNE A, #0FH, COL1       ; 若行有按下按鍵，跳至列偵測

ROW2:
    MOV P1, #0BFH            ; 掃描第 2 列（P1.6 = 0）
    LCALL DELAY
    MOV A, P1
    ANL A, #0FH
    MOV R0, #4               ; 記錄第 2 列 base index 為 4
    CJNE A, #0FH, COL1       ; 若有按鍵，跳至列偵測
    JMP ROW1                 ; 無按鍵，回到列 1 繼續掃描

;======================
; 行偵測並記錄編號
;======================
COL1:
    CJNE A, #0EH, COL2
    MOV R1, #0               ; 第 1 行 → R1 = 0
    JMP DO
COL2:
    CJNE A, #0DH, COL3
    MOV R1, #1               ; 第 2 行 → R1 = 1
    JMP DO
COL3:
    CJNE A, #0BH, COL4
    MOV R1, #2               ; 第 3 行 → R1 = 2
    JMP DO
COL4:
    CJNE A, #07H, ROW1
    MOV R1, #3               ; 第 4 行 → R1 = 3
    JMP DO

;======================
; 音符判斷處理（按鍵編號 = R0 + R1）
;======================
DO:
    MOV A, R0
    ADD A, R1                ; A = 按鍵編號（0~7）

    CJNE A, #0, RE
```

```
    ACALL PLAY_DO
    AJMP ROW1

RE:
    CJNE A, #1, MI
    ACALL PLAY_RE
    AJMP ROW1

MI:
    CJNE A, #2, FA
    ACALL PLAY_MI
    AJMP ROW1

FA:
    CJNE A, #3, SO
    ACALL PLAY_FA
    AJMP ROW1

SO:
    CJNE A, #4, LA
    ACALL PLAY_SO
    AJMP ROW1

LA:
    CJNE A, #5, SI
    ACALL PLAY_LA
    AJMP ROW1

SI:
    CJNE A, #6, SONG
    ACALL PLAY_SI
    AJMP ROW1

;======================
; 播放小蜜蜂旋律（A = 7 時）
;======================
SONG:
    MOV DPTR, #SONG_TABLE
    MOV R7, #14                ; 播放 14 個音符

SONG_LOOP:
```

```
    CLR A
    MOVC A, @A+DPTR          ; 取出當前音符編號
    INC DPTR
    CJNE A, #255, SONG_DISPATCH
    LJMP ROW1                ; 若遇 255（終止符號），回掃描主程式

;======================
; 音符對應播放（查表分派）
;======================
SONG_DISPATCH:
    CJNE A, #0, SONG_RE
    ACALL PLAY_DO
    AJMP SONG_NEXT
SONG_RE:
    CJNE A, #1, SONG_MI
    ACALL PLAY_RE
    AJMP SONG_NEXT
SONG_MI:
    CJNE A, #2, SONG_FA
    ACALL PLAY_MI
    AJMP SONG_NEXT
SONG_FA:
    CJNE A, #3, SONG_SO
    ACALL PLAY_FA
    AJMP SONG_NEXT
SONG_SO:
    CJNE A, #4, SONG_LA
    ACALL PLAY_SO
    AJMP SONG_NEXT
SONG_LA:
    CJNE A, #5, SONG_SI
    ACALL PLAY_LA
    AJMP SONG_NEXT
SONG_SI:
    ACALL PLAY_SI            ; A = 6 預設為 Si

SONG_NEXT:
    ACALL DELAY
    DJNZ R7, SONG_LOOP
    LJMP ROW1                ; 播完回主程式
```

```
;======================
; 小蜜蜂旋律音符編碼表
;======================
SONG_TABLE:
    DB 0,0,4,4,5,5,4        ; Do Do So So La La So
    DB 3,3,2,2,1,1,0        ; Fa Fa Mi Mi Re Re Do
    DB 255                  ; 255 = 播放結束標記

;======================
; 各音符播放子程式
;======================
;======================
; PLAY_DO - 播放 Do 音符
;======================
PLAY_DO:
    MOV TL0, #12            ; 設定 Timer0 低位，對應 Do 的頻率
    MOV TH0, #196           ; 設定 Timer0 高位（TH0:TL0 = C4CH），產生約 523Hz 的方波頻率

    MOV R3, #3              ; 外層迴圈次數，控制音符的總時長（此值可調整音長）
    SETB TR0                ; 啟動 Timer0 計數

PDLOOP:
    MOV R2, #130            ; 內層迴圈次數，每次呼叫 INVERTDO 來翻轉一次蜂鳴器輸出

PDLOOP1:
    ACALL INVERTDO          ; 呼叫產生方波的副程式
    DJNZ R2, PDLOOP1        ; 內圈次數減一，沒減完則繼續
    DJNZ R3, PDLOOP         ; 外圈次數減一，沒減完則整個再來一次

    CLR TR0                 ; 播放完成，關閉 Timer0
    RET                     ; 返回主程式


PLAY_RE:
    MOV TL0, #28
    MOV TH0, #202
    MOV R3, #3
    SETB TR0
PRLOOP: MOV R2, #196
PRLOOP1: ACALL INVERTRE
```

```
        DJNZ R2, PRLOOP1
        DJNZ R3, PRLOOP
        CLR TR0
        RET

PLAY_MI:
    MOV TL0, #21
    MOV TH0, #208
    MOV R3, #3
    SETB TR0
PMLOOP: MOV R2, #220
PMLOOP1: ACALL INVERTMI
        DJNZ R2, PMLOOP1
        DJNZ R3, PMLOOP
        CLR TR0
        RET

PLAY_FA:
    MOV TL0, #8
    MOV TH0, #211
    MOV R3, #3
    SETB TR0
PFLOOP: MOV R2, #233
PFLOOP1: ACALL INVERTFA
        DJNZ R2, PFLOOP1
        DJNZ R3, PFLOOP
        CLR TR0
        RET

PLAY_SO:
    MOV TL0, #5
    MOV TH0, #216
    MOV R3, #4
    SETB TR0
PSLOOP: MOV R2, #196
PSLOOP1: ACALL INVERTSO
        DJNZ R2, PSLOOP1
        DJNZ R3, PSLOOP
        CLR TR0
        RET
```

```
PLAY_LA:
    MOV TL0, #16
    MOV TH0, #220
    MOV R3, #4
    SETB TR0
PLLOOP: MOV R2, #220
PLLOOP1: ACALL INVERTLA
        DJNZ R2, PLLOOP1
        DJNZ R3, PLLOOP
        CLR TR0
        RET

PLAY_SI:
    MOV TL0, #12
    MOV TH0, #224
    MOV R3, #4
    SETB TR0
PSILOOP: MOV R2, #247
PSILOOP1: ACALL INVERTSI
         DJNZ R2, PSILOOP1
         DJNZ R3, PSILOOP
         CLR TR0
         RET

;======================
; 音符對應方波產生副程式（Timer0 + CPL）
;======================
;======================
; INVERTDO - 反轉蜂鳴器輸出腳並等待 Timer0 溢位
;======================
INVERTDO:
    CPL P0.0              ; 翻轉 P0.0 腳位（蜂鳴器輸出）→ 產生方波的一半週期
    JNB TF0, $            ; 等待 Timer0 溢位（Timer 溢位旗標 TF0 = 1）
    CLR TF0               ; 清除 TF0 溢位旗標，以便下次計數
    MOV TL0, #12          ; 重設 Timer0 初值（TL0）
    MOV TH0, #196         ; 重設 Timer0 初值（TH0）→ 確保下一輪計時
    RET                   ; 返回呼叫者（如 PLAY_DO）


INVERTRE: CPL P0.0
    JNB TF0, $
```

```
        CLR TF0
        MOV TL0, #28
        MOV TH0, #202
        RET

INVERTMI: CPL P0.0
        JNB TF0, $
        CLR TF0
        MOV TL0, #21
        MOV TH0, #208
        RET

INVERTFA: CPL P0.0
        JNB TF0, $
        CLR TF0
        MOV TL0, #8
        MOV TH0, #211
        RET

INVERTSO: CPL P0.0
        JNB TF0, $
        CLR TF0
        MOV TL0, #5
        MOV TH0, #216
        RET

INVERTLA: CPL P0.0
        JNB TF0, $
        CLR TF0
        MOV TL0, #16
        MOV TH0, #220
        RET

INVERTSI: CPL P0.0
        JNB TF0, $
        CLR TF0
        MOV TL0, #12
        MOV TH0, #224
        RET

;=====================
```

```
; 鍵盤掃描與音符播放之間的延遲
;=====================
DELAY:
    MOV R5, #100
DELAY1:
    MOV R6, #100
DELAY2:
    DJNZ R6, DELAY2
    DJNZ R5, DELAY1
    RET

    END
```

六、心得：

上課心得:

在本次課程中，複習到 8051 微控制器中 Timer 的基本運作模式及其實際應用，老師講解如何利用 Timer 產生精確延遲與週期訊號時，我開始理解計時器並不僅是延遲用途，更能夠被廣泛應用在控制頻率與節奏的場景中，例如音頻控制與脈波輸出。

透過實際推導 Timer 計數公式與 THx / TLx 的設定方式，我更清楚掌握了頻率與 Timer 值之間的對應關係，這為後續實驗中播放音階打下了重要基礎。

實驗心得:

在本次實驗中，我利用矩陣鍵盤作為輸入裝置，透過偵測不同按鍵來播放對應的音符（Do 到 Si）。而實現音符播放的核心技術，就是透過 8051 的 Timer0 搭配 CPL 指令產生特定頻率的方波，進而驅動蜂鳴器發聲。

我將 Timer0 設定為 Mode 0，指定 TH0 與 TL0 的初值，使 Timer 每次溢位的時間精確控制在音符所需的週期內。並透過多次的 CPL 翻轉與 Timer 重載，產生穩定的方波音頻，並配合內外迴圈控制音長，讓整個音符的音高與時值都可由程式邏輯決定，我深刻體會到 Timer 不只是計時工具，更是一個可以控制輸出波形頻率的關鍵元件。

此外，進階加分題讓我嘗試以查表方式播放完整旋律（如小蜜蜂、小星星），進一步熟悉了如何結合程式邏輯、Timer 控制與輸出裝置達成完整的應用效果，實作過程讓我收穫良多。

Notes:

1. 內容字體大小為 12，間距為單行間距
2. 中文字字體為標楷體
3. 英文字和阿拉伯數字為 Times New Roman
4. 嚴禁抄襲，抄襲者以 0 分計算
5. 請於報告左上角附上照片
6. 每次實驗課繳交上次實驗結報